Math 2130: Calculus 3 (10/28/2019)

Computing a volume in Maple.

```
> restart;
  with(plots):
```

Find the volume of the solid bounded by z = 9 - x^2 - y^2 and z = -2 where also x >= 0.

We started with graphs of z = 9 - x^2 - y^2 and z = -2 plotted over domains of x and y going from -3 to 3. Then we used the bounds and their intersections to fix domains until we got the plot below.

```
> A := plot3d(9-x^2-y^2,x=0..sqrt(11-y^2),y=-sqrt(11)..sqrt(11),color=
  blue):
  B := plot3d(-2,x=0..sqrt(11-y^2),y=-sqrt(11)..sqrt(11),color=red):
  C := plot3d([0,y,z],y=-sqrt(11)..sqrt(11),z=-2..9-y^2,color=green):

  display({A,B,C},scaling=constrained);
```



We noted that the above plots actually tell us how to set up the integral in the order...

dz dx dy

```
> int(int(int(1,z=-2..9-x^2-y^2),x=0..sqrt(11-y^2)),y=-sqrt(11)..sqrt
  (11));
```

$$\frac{121\,\pi}{4} \qquad (1)$$

Adjusting to the next order was straight forward (we consider the half disk as y simple instead of x simple)...

dz dy dx

```
> int(int(int(1,z=-2..9-x^2-y^2),y=-sqrt(11-x^2)..sqrt(11-x^2)),x=0..
  sqrt(11));
```

$$\frac{121\,\pi}{4} \qquad (2)$$

The last order we considered in class "required" us to solve some equations. For example, the blue surface gave us both y bounds (the "right" and "left" hand side of the solid), so we took the blue surface's equation and solved for y to get those bounds. Other bounds were either given or came from intersecting previous bounds.

dy dx dz

```
> solve(z=9-x^2-y^2,y);
```

$$\sqrt{-x^2 - z + 9},\ -\sqrt{-x^2 - z + 9} \quad (3)$$

```
> solve(sqrt(-x^2-z+9)=-sqrt(-x^2-z+9),x);
```

$$\sqrt{-z + 9},\ -\sqrt{-z + 9} \quad (4)$$

```
> int(int(int(1,y=-sqrt(9-x^2-z)..sqrt(9-x^2-z)),x=0..sqrt(9-z)),z=-2..9);
```

$$\frac{121\,\pi}{4} \quad (5)$$

As long as we're at it, let's do the other 3 orders of integration. I'll just write them down. You can look at them and decide were the bounds came from. :)

dy dz dx

```
> int(int(int(1,y=-sqrt(9-x^2-z)..sqrt(9-x^2-z)),z=-2..9-x^2),x=0..sqrt(11));
```

$$\frac{121\,\pi}{4} \quad (6)$$

dx dz dy

```
> int(int(int(1,x=0..sqrt(9-y^2-z)),z=-2..9-y^2),y=-sqrt(11)..sqrt(11));
```

$$\frac{121\,\pi}{4} \quad (7)$$

dx dy dz

```
> int(int(int(1,x=0..sqrt(9-y^2-z)),y=-sqrt(9-z)..sqrt(9-z)),z=-2..9);
```

$$\frac{121\,\pi}{4} \quad (8)$$

Finally, if evaluating this integral by hand, we wouldn't want to use any of these interated integrals. This integral has nice circular symmetry so we would want to set it up in cylindrical/polar coordinates.

```
> int(int(int(1*r,z=-2..9-r^2),r=0..sqrt(11)),theta=-Pi/2..Pi/2);
```

$$\frac{121\,\pi}{4} \quad (9)$$